
ABSTRACT  

Determining enrollment target and projecting realistic enrollment rates are challenging in conducting a clinical trial. 
Underperforming participant recruitment has been a long-time problem for clinical trials. To address the question 
whether the ongoing recruitment will enable a trial to reach the planned target number fast enough while staying 
within budget, the real-time monitoring is key to a successful trial. This paper shows how to use SAS® to 
automatically visualize the dynamic accrual progression, and simultaneously make a data-driven projection on the 
accrual rate and the completion date of enrollment target. This real data driven projection can work as a reference for 
the study team and sponsors to make necessary future amendment to enrollment planning and timeline scheduling.  

 

INTRODUCTION  

Determining and projecting realistic enrollment rates is one of the most challenging aspects of conducting a clinical 
trial. Ninety percent of worldwide clinical trials were delayed due to under-performing participant recruitment. Study 
timelines are typically extended to nearly double their original duration to meet desired enrollment levels for all 
therapeutic areas. Among the reasons that delay the recruitment, the optimism bias in clinical trial recruitment can 
lead to underutilized staff, which is always the biggest expense of a clinical trial. For grant-funded research, 
recruitment delays can jeopardize the research effort, the failure to reach target sample size post the run-out-of-
funding would result in amending the eligibility criteria to meet the enrollment goal. Another “wait-and-see” approach 
occurs to researchers although they are aware of that 90% of clinical trials are delayed due to recruitment but they 
assume that their clinical trial would fall into the other 10% of trials not to be delayed in recruitment.  

To proactively prevent recruitment disaster that might result from overestimation based on intuition, it is critical for 
research institute to both build effective pre-recruitment strategies and keep closely monitoring the progress in 
participants recruitment and the rate of participants drop-out due to exclusion or early termination at each stage of the 
typical enrollment. The derived real time data driven projection on enrollment rate and completion date of enrollment 
target make it possible to fit the real-time enrollment funnel model and make appropriate updates in the enrollment 
plan to meet the final enrollment goal.  

SAS® procedures and MACRO facility make it easy to automatically report the recruitment/enrollment data and 

visualize the progress of study accrual to help the study team get updated timely on the issues occurred in screening, 
randomization/intervention, subject-specific protocol deviation and early termination, which may have the negative 
impact on a successful fulfillment of the enrollment goal. The SAS/STAT procedures are useful tools to help generate 
the projected target completion date and forthcoming rate in trial enrollment.   

In this paper we present examples using SAS® to keep track of a multi-site clinical trial study accrual by real-time 

monitoring raw eCRF data and to make the projection on study enrollment target completion, which can be used as 
the reference for sponsors to identify and prevent possible under-enrollment in the trial. 

 

CASE STUDY 1: USING METRICS TO MONITOR STUDY ACCRUAL 

This section shows using SAS® to track participants enrollment.  

INTRODUCTION TO CLINICAL TRIAL WORKFLOW 

Figure 1 is the representation of a typical clinical trial enrollment process once the clinical site is activated with the 
determined maximum number of participants. The screening process evaluates potential participants for their 
eligibility to take part in a study. The enrollment and randomization processes register participants into a clinical trial, 
assigning them to the appropriate treatment arm. Once participants are enrolled, they can receive the assigned study 
product or intervention  

Figure 1. Schematic Illustration of Typical Clinical Trial Workflow Post Site Activation  

 

METRICS REPORT TO REPRESETN ENROLLMENT ACCRUAL 

The Case Report Forms (CRFs or eCRFs) such as Screening Failure Checklist (SFC), Enroll (ENR), Demographic 
(DEM), Consent Agreement (CS), Visit Documentation (VD) are to be used to retrieve the real time data for Trial 
Metrics Report. As shown in Figure 2 the example trial accrual report to monitor study accrual at each individual trial 
site as well as the allover accrual, the study team can keep updated with the accumulation in screening failures, 
average enrollment ready for randomization each week. To make SAS® programming more efficient, a derived 
comprehensive dataset has been generated and used for programming, which includes all useful variables from 
multiple different types of eCRF, such as participant’s ID, start date of screening,  date when the participant was 
determined to be a screening failure, whether to meet the eligibility criteria, and what this criterion is, etc.  

 

Table 1. Metrics Report – Study Accrual by Site  

 

GRAPHS TO REPRESENT ENROLLMENT FUNNEL 

In addition, by creating a dataset with the average drop-out rate for each of the different stages of enrollment, we can 
generate the graph to represent the enrollment funnel for randomized participants at each site. Based on the example 
graph shown in Figure 3, Site A and B need to screen about 150 participants for ~70 enrollments ready for 
randomization, whereas Site C and D demonstrate a lower drop-out rate. By comparing the graphs of Enrollment 
Funnel, it makes it easy to compare the performance of different site in trial recruitment.  

In the case of clinical trial study of different protocols, the real data driven enrollment funnel can help determine 
whether the trial is an average or difficult study regarding participants recruitment, but also forecast how many 
recruitments would need at top of the recruitment funnel in order to hit the enrollment/randomization goal.  

 

 

 

 

 

 

 

Figure 2. Enrollment Funnel for each site or overall 

EXAMPLE DATASET AND SAS PROGRAM TO GENERATE THE GRAPH OF 
ENROLLMENT FUNNEL 

Following is an example dataset read in by the example SAS® program used to generate the Enrollment Funnel. 

 

proc format; 

     value catf 0="Total Recruitment" 

                1="Screen Passed" 

                2="Randomized" 

                3="Treated"; 

run; 

%macro enrfunnel(dsn); 

   proc sgplot data=&dsn.; 

        format cat catf.; 

        hbarparm category=CAT response=COUNT  

               / dataskin=Pressed datalabel=txt 

          datalabelpos=data 

          datalabelattrs=(color=darkgreen Family="Arial" size=14 Weight=bold  

style=normal) 

                barwidth=0.7; 

        xaxis display=(nolabel) offsetmin=0 offsetmax=0.15  

           valueattrs=(size=14 weight=Bold color=blue)values=(0 to 200 by 100); 

        yaxis display=(nolabel) offsetmax=0.15 offsetmin=0.15  

           valueattrs=(size=14 weight=Bold color=blue); 

   run; 

%mend enrfunnel; 

%enrfunnel(temp) 

CASE STUDY 2: PROJECTION ON COMPLETION DATE TO MEET ENROLLMENT 
TARGET 

This section introduces SAS® program to generate graphs to keep track of the progress in randomization and the 

projected data driven enrollment completion date.  

This SAS® program is composed of three components, as shown in Figure 4. The first macro is to use FREQ 
procedure to get the cumulative enrollment by each group and date. The second macro is using GLM procedure (as 
an example) to predict the date to meet the required enrollment target for randomization. The third macro is to use 
SGPLOT procedure to generate the graph. Figure 5 is an example output to visualize the enrollment progress and 
the estimated completion date, 05JUL2022, based on the average enrollment observed More details can be found in 
the following SAS® program with an example dataset included.  

 

 

Figure 3. Flowchart of SAS® program to track study accrual and to make projection on enrollment 
target completion date 

Figure 4. Enrollment Progress and Projection on the Target Completion Date  

EXAMPLE CODE TO GENEREATE GRAPH OF THE ENROLLMENT PROGRESS MARKED 
BY PROJECTED ENROLLMENT COMPLETION DATE TO MEET THE ENROLLMENT 
TARGET 

Following is an example dataset Temp read in by the example SAS® program 

 

/*PART 1: Add new variable _grp.  

          _grp="projected" when CUM=50, the Target Enrollment */ 

proc sort data=temp;  

     by grp STARTDT;  

run; 

data temp1; 

     format _grp $9.; 

     set temp; 

  by grp STARTDT; 

  _grp="observed"; 

  output; 

  if last.grp then do; 

     STARTDT=.; 

  COUNT=.; 

  CUM=50; 

  _grp="projected"; 

  output; 

  end; 

run; 

/*PART 2: Use GLM to fit the Model Stardt = CUM.  

        Predicted value is the projected completion data for the Enrollment Target, says CUM=50*/ 

ods output ParameterEstimates=pars; 

Proc GLM data=temp1; 

 Model startdt=cum; 

 output out=temp2 predicted=fit; 

run;  

quit; 

ods output close; 

data temp3; 

     format _grp $9. fit date9.; 

     set temp2; 

     _grp="projected"; 

     startdt=fit; 

     output; 

     keep _grp grp grpc STARTDT CUM; 

run; 

/*PART 3: Combine two datasets _grp="observed" and _grp="projected". Output the graph. */ 

data temp4; 

     set temp2(where=(_grp="observed"))  

         temp3(rename=(CUM=pCUM)); 

run; 

proc sql noprint; 

     /*To assign the predicted completion date to macro variables */ 

     select STARTDT format=f5. into: projdt_1  trimmed   

            from temp4  

            where _grp in("projected") and grp=1 and pCUM=50;  

     select STARTDT format=date9. into: projdt_1c  trimmed   

            from temp4  

            where _grp in("projected") and grp=1 and  pCUM=50;  

 

     select min(STARTDT) format=f5.0 into: min_dt  trimmed  from temp4; 

     select max(STARTDT) format=f5.0 into: max_dt  trimmed  from temp4; 

quit; 

ods graphics on; 

proc sgplot data=temp4; 

     label pCUM="Projected CUM"; 

     series  x=STARTDT y=pCUM /lineattrs=(thickness=3 color=blue   pattern=solid); 

     scatter x=STARTDT y=CUM  /markerattrs=(size=13 color=green  symbol=circle); 

     refline 50/axis=y LINEATTRS=(pattern=shortDash color=green ) label=('Group 1 Target=50') 

                labelattrs=(size=8 weight=bold family=arial)  

                LABELLOC = outside LABELPOS=MAX splitchar="*"; 

     refline &projdt_1/axis=x LINEATTRS=(pattern=shortDash color=blue ) label=("&projdt_1c")  

                labelattrs=(size=8 weight=bold family=arial color=blue)  

                LABELLOC = OUTSIDE LABELPOS=max; 

     xaxis display=(nolabel) VALUES= (&min_dt to %eval(&max_dt+30) by 7) grid; 

     yaxis label = 'Cumulative Number Randomized'   values=(0 to 80 by 20) grid;  

run; 

ods graphics off; 

 

CONCLUSION  

Underperforming participant recruitment is a long-time problem for clinical trials.  Determining enrollment 
target and projecting realistic enrollment rates are challenging in conducting a clinical trial. In this paper 
we present examples using SAS® to keep track of study accrual by generating the metrics report, 
enrollment funnel graph, and making projection on the date to meet study enrollment target, to help 
sponsors identify and prevent under-enrollment.  

